

Constructing a Möbius Band from a Rectangular Strip

An Exercise of Maple Graphics

Carl Eberhart
Department of Mathematics
University of Kentucky
Lexington, Kentucky

September 7, 1994

Statement of the problem

Take a long strip of paper P , fasten the ends together after putting a twist in it and you have made a Möbius Band M , a well-known example of a *non orientable surface*. Don Buckholtz, a colleague in the Mathematics Department, has posed the problem of finding equations which carry out this construction in a 'natural manner', that is, one in which no distortions of the surface occur which would cause the paper to tear or even stretch. Don came up with this problem while teaching Calculus III using Maple, and produced some ingenious equations for winding an S shaped piece of paper onto a Möbius Band on a pair of carefully positioned cones. In this note, we will give two other solutions to the problem.

A fake solution.

It is easy to find transformations which model the construction if we ignore the 'naturalness' constraint. For example, think of a vertical stick $2h$ units high whose center at time t is located on the xy -plane at $(r \cos \theta, r \sin \theta, 0)$. Now rotate the center of the stick counterclockwise around the origin in the xy -plane, and while you are at it, rotate the stick around its center in the revolving plane spanned by $N = [0, 0, 1]$ and $K = [-\sin \theta, \cos \theta, 0]$. Perform this rotation at half speed, so that the stick rotates π radians about its center as θ goes from 0 to 2π . As it rotates, the stick will sweep out a Möbius Band.

To accomplish this in Maple, first set up a moving basis of orthonormal vectors which we can use to express an arbitrary point on the band.

```
> J := [cos(theta), sin(theta), 0];  
> K := [-sin(theta), cos(theta), 0];  
> N := [0, 0, 1];
```

$$J := [\cos(\theta), \sin(\theta), 0]$$

$$K := [-\sin(\theta), \cos(\theta), 0]$$

$$N := [0, 0, 1]$$

Next set the radius r and height h of the band at say

```
> r := 3: h := 1:
```

Now any point w on the band can be expressed as a function F of t between $-h/2$ and $h/2$ and θ between 0 and 2π .

```
> w := evalm(r*j+t*(cos(theta/2)*K+sin(theta/2)*N)):
> F := unapply([w[1],w[2],w[3]],(theta,t));
```

```
F := (theta,t) ->
```

$$\left[r \cos(\theta) - t \cos\left(\frac{1}{2}\theta\right) \sin(\theta), r \sin(\theta) + t \cos\left(\frac{1}{2}\theta\right) \cos(\theta), t \sin\left(\frac{1}{2}\theta\right) \right]$$

Drawing a picture of the range of the transformation $F : [0, 2\pi] \times [-h/2, h/2] \rightarrow R^3$.

```
> plot3d(F(theta,t),theta=0..2*Pi,t=-h/2..h/2,style=patch,orientation=[30,68],
> shading=zgrayscale,scaling=constrained);
```



2in4in

The problem with this transformation is that it does not realistically model the actual band that is constructed from a strip of paper. A realistic transformation doing this would need to have the property that a curve drawn on the strip would have the same length as its transform, and F does not preserve length. For example the top of the strip has length $2\pi r$, but its transform has a length which depends on the height h of the strip. To see this, First get the element of arc length as a function of θ and make a simplifying substitution.

```
> dl := map(diff,F(theta,h/2),theta):
> dl2 := 4*(simplify(expand(subs(theta=2*alpha,linalg[dotprod](dl,dl)),trig),trig));
```

$$dl2 := h^2 \cos(\alpha)^2 + 4r^2 - 2h \sin(\alpha)r + \frac{1}{4}h^2$$

Now integrate from 0 to π to get the length.

```
> tlen := int(sqrt(dl2),alpha=0..Pi);
```

$$tlen := \int_0^\pi \frac{1}{2} \sqrt{4h^2 \cos(\alpha)^2 + 16r^2 - 8h \sin(\alpha)r + h^2} d\alpha$$

Maple gives up on computing this integral symbolically, but we can get a numerical value easily for say a radius of 3 units and a height of 1 unit.

```
> evalf(subs({r=3,h=1},tlen));
```

18.01812221

and note that this is less than the original length of

```
> evalf(2*3*Pi);
```

18.84955592

A solution by folding.

On the face of it, it ought to be possible to modify the transformation F slightly to obtain one which is length preserving. I have worked on this approach, but my knowledge of physics is insufficient to let me find a solution. While reading the chapter entitled Shortest Möbius Band in Steven Barr's *Experiments in Topology*, I got the idea of folding the strip to form the Möbius Band. This is a

Programming problem:

Given a strip in the xy plane, sequences of points B_0, B_1, \dots, B_n and A_0, A_1, \dots, A_n along the top and bottom of the strip respectively, and a sequence $\theta_1, \theta_2, \dots, \theta_{n-1}$ of angles we want to fold the strip through about the segments connecting the corresponding points on the top and bottom sequences, design a Maple word to carry this out and draw the final picture of the folded strip.

A Solution to the programming problem:

First, we need a function which takes an angle θ and a quadrilateral B_0, A_0, A_1, B_1 and returns the rotation of 3-space about the line through A_1 and B_1 by an angle θ . Specifically, let J be the unit vector in the direction from A_1 to B_1 , Let N be the unit vector pointing up in the direction of $A_1 - A_0$ crossed with $B_0 - A_0$, and let K be the unit vector J cross N . In the plane. Then for an arbitrary point X , its image under this rotation is obtained by the following sequence of steps:

- Form $Y = X - A_1$, the vector from A_1 to X .
- Form $P = [(Y \cdot K), (Y \cdot N)]$, the orthogonal projection of Y onto the plane spanned by N and K .
- Rotate P by θ radians in this plane to the point Q .
- Project Q to the plane spanned by N and K , then add to this A_1 and the projection of Y to the line spanned by J .

The Maple procedure T below constructs the transformation just described. It is used by the procedure `fold` below.

```
> T := proc(th,B0,A0,A1,B1)
> local X,Y,J,N,K,P,m;
> X := linalg[vector](3);
> evalm(linalg[crossprod](A1-A0,B0-A0));
> N:= evalm("/sqrt(linalg[dotprod](N,N))");
> J := evalm((B1-A1)/sqrt(linalg[dotprod](B1-A1,B1-A1)));
> K := evalm(linalg[crossprod](J,N));
> Y := evalm(X-A1);
> P :=linalg[vector]([linalg[dotprod](Y,K),linalg[dotprod](Y,N)]);
> Q := evalm(linalg[matrix]([cos(th),-sin(th),
>                               [sin(th),cos(th)]])*P);
> evalm(A1+linalg[dotprod](Y,J)*J + Q[1]*K +Q[2]*N);
> convert(",list");
> unapply(",X)
> end;
```

Maple V3 has a word, `polygon3dplot`, in its `plots` package which takes a sequence of points in 3-space and draws the polygon with these points as vertices. We will use `polygon3dplot` to draw

our Möbius Band as a sequence of quadrilaterals.

Now we need a function which takes the list of bottom points and top points and returns a set of quadrilaterals suitable for input to `polygon3dplot`.

It, like the function `T` above, will be used by the procedure `fold` below.

```
> mkpoly := proc(bot,top)
>   local l,i;
>   l := NULL;
>   for i from 1 to nops(bot)-1 do
>     l := l,[top[i],bot[i],bot[i+1],top[i+1]] od; l end;
```

Now the idea of `fold` is as follows:

- First take the sequences of top B_0, B_1, \dots, B_n , and bottom points A_0, A_1, \dots, A_n , and produce a sequence of quadrilaterals $Q_1 = [B_0, A_0, A_1, B_1]$, $Q_2 = [B_1, A_1, A_2, B_2]$, \dots , $Q(n) = [B(n-1), A(n-1), A(n), B(n)]$. Use `mkpoly` to do this.
- Next use `T` to get the rotation of space about the line through A_1 and B_1 by the angle θ_1
- Apply the rotation to each of the quadrilaterals Q_2, Q_3, \dots, Q_n . obtaining quadrilaterals Q_2', Q_3', \dots, Q_n' .
- Next go back and use `T` again to get the rotation of space about the line through A_2' and B_2' by the angle θ_2 .
- As before apply the rotation to each of the quadrilaterals Q_3', \dots, Q_n' .
- Continue this pattern until the all angles have been used.

It is easy to see that resulting folded strip is obtained from the original rectangular strip in a manner that does not distort by stretching or tearing the paper. Each of the quadrilaterals in the original strip is carried by a sequence of rotations to a quadrilateral on the folded strip. So there is no stretching. Further, adjacent quadrilaterals are carried to adjacent quadrilaterals, so there is no tearing.

To implement this in Maple, we can proceed as follows.

```
> fold := proc(bot,top,ang)
>   local k,st,p,n,i;
>   n := nops(bot);
>   st := mkpoly(bot,top);
>   for i from 1 to n-2 do
>     p := T(ang[i],op(st[i]));
>     st := seq(st[k],k=1..i),seq(map(p,st[k]),k=i+1..n-1);
>     od;
>     [st];
>   end;
```


We test our procedures out by folding a Möbius Band. This can be done with a sequence of six right angle folds on a strip 1 unit high and a carefully chosen $25 - 3\sqrt{2}$ units long.

```
> ang := [Pi/2,Pi/2,Pi/2,Pi/2,Pi/2,Pi/2]:
> top := [[0,1,0],[2,1,0],[4,1,0],[8,1,0],[9,1,0],
>         [14,1,0],[20,1,0],[25-3*sqrt(2),1,0]]:
> bot := [[0,0,0],[1,0,0],[5,0,0],[7,0,0],[9,0,0],[14,0,0],
>         [20,0,0],[25-3*sqrt(2),0,0]]:
```

```

> strip := fold(bot,top,ang):
> plots[polygonplot3d]({strip},scaling=unconstrained,axes=none,
>     orientation=[-146,47],shading=zgrayscale);

```



3in4in

A solution by bending.

Now, the Möbius Band you get by twisting and taping a strip of paper is not the Band we have described above. For one thing, that transformation is *smooth*, that is, each point on the Band has a tangent plane to the Band. How can we obtain a smooth transformation?

We propose to 'round off' the corners by bending our strip around some carefully positioned cylinders.

First, we need a procedure to 'back off' or 'snip out' the folds by ϵ on both sides. This amounts to constructing another set of smaller quadrilaterals and transforming them to form a folded strip with the folds cut away. The procedures `mkpoly2` and `fold2` described below are used in performing this construction.

```

> mkpoly2 := proc(bot,top,eps)
>   local l,i;
>   l := NULL;
>   for i from 1 to nops(bot)-1 do
>     l := l,[[top[i][1]+eps,top[i][2..3]],bot[i][1]+eps,bot[i][2..3]],
>         [bot[i+1][1]-eps,bot[i+1][2..3]],top[i+1][1]-eps,
>         top[i+1][2..3]] ] od;
>   l
> end:

```

Next, we define a procedure `fold2`, which generates the folded strip of quadrilaterals with the folds cut away.

```

to polygonplot3d.
> fold2 := proc(bot,top,ang,eps)
>   local k,st,p,n,i,st2;
>   n := nops(bot);
>   st := mkpoly(bot,top);
>   st2 := mkpoly2(bot,top,eps);
>   for i from 1 to n-2 do
>     p := T(ang[i],op(st[i]));
>     st := seq(st[k],k=1..i),seq(map(p,st[k]),k=i+1..n-1);

```

```

> st2 := seq(st2[k],k=1..i),seq(map(p,st2[k]),k=i+1..n-1);
> od;
> [st2]
> end:


```

Checking this out on our Möbius band -

```

> strip2 := fold2(bot,top,ang,1/5):
> pl2 := plots[polygonplot3d]({strip2}):
> plots[display3d]({pl2},axes=none,orientation=[-146,47],shading=zgrayscale);

```



3in4in

Finally, we need a procedure `bend` which connects the backed off polygons with strips wound onto certain carefully positioned cylinders with carefully chosen radii (called `ra` in the procedure).

```

> bend := proc(eps,th,B0,A0,A1,B1)
>   local J,N,K,ra,beta,lamda,l,m;
>   evalm(linalg[crossprod](A1-A0,B0-A0));
>   N:= evalm("/sqrt(linalg[dotprod](",","))");
>   l := sqrt(linalg[dotprod](B1-A1,B1-A1));
>   J := evalm((B1-A1)/l);
>   K := evalm(linalg[crossprod](J,N));
>   beta := linalg[angle](evalm(B1-A1),evalm(A1-A0));
>   lamda := eps*sin(beta);
>   ra := lamda*tan((Pi-abs(th))/2);
>   if evalf(th) > 0 then
>     evalm(A1+ra*N+t*(B1-A1)+ra*(cos(al)*(-N)+sin(al)*K)+
>       (al/th*2*eps*cos(beta))*J);
>     plot3d(["[1]","[2]","[3]"],t=0..1,al=0..th,style=wireframe):
>   else
>     evalm(A1-ra*N+t*(B1-A1)+ra*(cos(Pi/2-al)*(K)+sin(Pi/2-al)*N)+
>       (al/(-th)*2*eps*cos(beta))*J);
>     plot3d(["[1]","[2]","[3]"],t=0..1,al=0..-th,
>       color=gray,style=patch) fi:
> end:

```

Now we generate the 6 bends needed for the band. Actually there are 7 needed, but the last one I omit so you can see where the strip begins and ends.

```

> for i from 1 to nops(strip2)-1 do
>   par.i := bend(1/5,ang(1)[i],op(strip2[i])) od:

```


Now generate a picture.

```

> plots[display3d]({pl2,seq(par.i,i=1..nops(strip2)-1)},axes=none,

```

```
> orientation=[-146,47],shading=zgrayscale);
```



3in4in

References

- [1] Barr, Stephen, *Experiments in Topology*, Dover Books, New York, 1989.